

## Count between dates by age range

Formula bar: `=COUNTIFS(joined,">="&start,joined,"<="&end,age,">="&LEFT(G8,FIND`

	A	B	C	D	E	F	G	H	I	J	K
1											
2		<b>Count between dates by age range</b>									
3											
4		<b>Name</b>	<b>City</b>	<b>Age</b>	<b>Joined</b>						
5		Bob	Cedar Rapids	29	1-Jan-20						
6		Mike	Chicago	77	25-Mar-20						
7		Rose	Des Moines	64	19-Apr-20						
8		Julie	Detroit	19	1-Jun-20						
9		Sam	New York	35	3-Jun-20						
10		Denise	Madison	63	18-Jul-20						
11		Kelly	Chicago	55	29-Jul-20						
12		Oliver	Sioux Falls	32	14-Aug-20						
13		Evilyn	St. Cloud	29	5-Sep-20						
14		Ellen	Ames	45	21-Sep-20						
15		Rose	Red Wing	21	2-Oct-20						
16		Lane	River Falls	24	4-Nov-20						
17											


<b>Start</b>	6/1/2020
<b>End</b>	10/1/2020
<b>Count</b>	7

0-20	1
21-30	1
31-50	3
51-70	2

*start = H4*  
*end = H5*  
*age = D5:D16*  
*joined = E5:E16*

**EXCELJET** 

## Summary

To count values between two dates that also fall into specific numeric ranges, you can use a formula based on the [COUNTIFS function](#), with help from the [LEFT](#), [RIGHT](#), [FIND](#), and [LEN](#) functions. In the example shown the formula in H8, copied down, is:

```
= COUNTIFS(joined, ">=" & start, joined, "<=" & end,  
age, ">=" & LEFT(G8, FIND("-", G8) - 1),  
age, "<=" & RIGHT(G8, LEN(G8) - FIND("-", G8)))
```

where **start** (H4), **end** (H5), **age** (D5:D16), and **joined** (E5:E16) are [named ranges](#). At each new row, the formula returns the count of all rows between start and end dates (inclusive) that also fall into the age range as shown in column G.

## Explanation

The goal of this example is to count rows in the data where the date joined falls between start and end dates (inclusive) and the age also falls into the age ranges seen in column G. The formula is complicated somewhat by the fact that the age range labels are actually text, so we need to extract a low and high number for each age range as a separate step.

Note: the named ranges shown in this example are entirely optional. They are a way to make the formula easier to enter, read, and copy.

## Count between dates

The main function used to solve this problem is [COUNTIFS](#). To explain how this works, we'll look first with at total seen in cell H6. This total *does not* take into account age groups, it simply counts all records that fall between the start and end dates. The formula in H6 is:

```
= COUNTIFS(joined, ">=" & start, joined, "<=" & end)
```

COUNTIFS is configured with two range/criteria pairs: one to count join dates greater than or equal to the start date in cell H4:

```
joined,">=" & start // greater than or equal to start
```

and one to count join dates less than or equal to the end date in cell H5:

```
joined,"<=" &end // less than or equal to end
```

Note that we are **concatenating** the operators inside the formula with the ampersand (&). The COUNTIFS function belongs to a **group of functions** that use this syntax.

With this configuration, COUNTIFS returns the total records with a join date greater than or equal to the start and end dates in H4 and H5.

### Count between age range

The formula above counts records using the start and end dates, but does not take into account age range. To further restrict the count to the age ranges shown in column G, we need to add two more range/criteria pairs. The first pair restricts the count to ages *greater than or equal to* the "low" number:

```
age, ">=" & LEFT(G8, FIND("-", G8) - 1) // low
```

Here, we use the FIND and LEFT function to extract the low number. The [FIND function](#) returns the position of the hyphen (-) and feeds this number (minus 1) to the [LEFT function](#) as the number of characters to extract. LEFT returns zero ("0"), which is concatenated to the greater than or equal to [operator](#) (>=). In the end, we have:

```
age, ">=0"
```

The second range/criteria pair restricts the count to ages *less than or equal to* the "high" number in the age range:

```
age, "<=" & RIGHT(G8, LEN(G8) - FIND("-", G8)) // high
```

As before, we use the FIND function to locate the position of the hyphen (-). The result is subtracted from the total of all characters in the cell (calculated with the [LEN function](#)) and this result is given to the [RIGHT function](#) for the number of characters to extract from the right side. RIGHT returns 20, which is concatenated to the less than or equal to operator (<=). In the end, we have:

```
age, "<=20"
```

As this formula is copied down the range H8:H11, the high and low values in the age ranges are extracted and used as conditions to restrict the count, while the original date logic remains unchanged.

## Author

Dave Bruns